IEOR 240 Case Study 1: Calgary Desk Company

Solar Shao, Xiaotong Guo, Hongfei Guo, Menghan Chen, Yuxuan Chen

1 Introduction

This report gives a recommended production schedule for CALDESCO's entire line of desks in September. From the perspective of the management team, how to achieve the maximum profit within the company's current production capacity and under practical constraints is often the most important and heavily discussed question. This report will discuss how to deliver a recommended production schedule for CALDESCO by maximizing its profit under all kinds of constraints.

CALDESCO manufactures student-size desks, standard desks, and executive desks in each of three lines, which are the economy line, basic pine line, and hand-crafted pine line. Hence, there would be 9 types of desk to be produced, which are "students desk in economy line", "students desk in basic pine line", "students desk in hand-crafted pine line", "standard desks in economy line", "standard desks in basic pine line", "standard desks in hand-crafted pine line", "executive desks in economy line", "executive desks in basic pine line", and "executive desks in hand-crafted pine line". Then the total profit of the company consists of the number of each type of desk produced times the profit of that type of desk.

On the other hand, the company's production activities must be carried out within its production capacity and under real-life constraints. Production constraints fall into four parts:

1.1 Material Availability

Aluminum, particle board, and pine sheets are materials used to produce different types of desks. The total consumption for production should not exceed the resource limitations.

1.2 Labor Hour and Production Line Availability

There are three production lines, and production time available on each of the production line are limited. The total production time of each production line occupied by the desk production should not exceed the specified operating time of the production line. Also, the total available labor hours are 230400 minutes, and the total amount of labor time required to produce a desk = 2 x total

production line time + hand crafting time + assembly time + finishing time. The total consumption for labor time should not exceed 230400 minutes.

1.3 Order Demand Constraints

The number of desks produced in September should be larger or equal to the amount ordered by clients.

1.4 Quota Restrictions

In order to sell all the desks it produces and to maintain its profit margins in part, CALDESCO sets quotas for each type of desk. The amount of each type of desk produced to the amount of all desks produced should conform to the range indicated by the quota.

2 Assumption

The following assumptions are made in this report:
1. By adhering to a set of in-house quotas, CALDESCO is able to sell all the desks it produces and maintains its profit margins.
2. Non-integer optimal solution is allowed. We assume that CALDESCO can still achieve a good profit without moving too far from the true optimal production goal by rounding down the optimal solution given by the LP relaxation, and we will prove this later in the analysis.
3. All profits and costs in this problem are in US dollars.
4. All labor-hour is measured in minutes.
5. All materials (aluminum, particle board, and pine sheets) are measured in square feet.

3 Formulation

The linear program is formulated based on the goal and limitations stated in the introduction.

3.1 Decision Variables

production amount$_{ESTU}$: amount of economy student desk produce
production amount$_{ESTA}$: amount of economy standard desk produce
production amount$_{EEXE}$: amount of economy executive desk produce
production amount$_{BSTU}$: amount of basic student desk produce
production amount$_{BSTA}$: amount of basic standard desk produce
production amount$_{BEXE}$: amount of basic executive desk produce
production amount$_{HCSTU}$: amount of hand crafted student desk produce

production amount$_{HCSTA}$: amount of hand crafted standard desk produce
production amount$_{HCEXE}$: amount of hand crafted executive desk produce

3.2 Objective Function

According to the previous assumption, CALDESCO is able to sell all the desks it produces by adhering to a set of in-house quotas. The goal is to maximize CALDESCO's profit in September within its production capacity and under real-life constraints. Hence, the objective function would be the sum of the number of each type of desk produced times the profit of that type of table:

$\max 20*\text{production amount}_{ESTU}+30*\text{production amount}_{ESTA}+40*\text{production amount}_{EEXE}$
$+50*\text{production amount}_{BSTU}+80*\text{production amount}_{BSTA}+125*\text{production amount}_{BEXE}+$
$100*\text{production amount}_{HCSTU}+250*\text{production amount}_{HCSTA}+325*\text{production amount}_{HCEXE}$

3.3 Constraints

3.3.1 Material Availability

The consumption of aluminum in square feet should not exceed the available amount 65000 square feet

$14*\text{production amount}_{ESTU}+24*\text{production amount}_{ESTA}+30*\text{production amount}_{EEXE} \leq$ 65000

The consumption of particle board in square feet should not exceed the available amount 60000 square feet

$8*\text{production amount}_{ESTU}+15*\text{production amount}_{ESTA}+24*\text{production amount}_{EEXE} \leq$ 60000

The consumption of pine sheets in square feet should not exceed the available amount 175000 square feet

$22*\text{production amount}_{ESTU}+40*\text{production amount}_{BSTA}+55*\text{production amount}_{BEXE}+$
$25*\text{production amount}_{HCSTU}+45*\text{production amount}_{HCSTA}+60*\text{production amount}_{HCEXE} \leq$ 175000

3.3.2 Labor Hour and Production Line Availability

Total production time (in minutes) of production line 1 should not exceed 9600 minutes

$1.5*\text{production amount}_{ESTU}+2*\text{production amount}_{ESTA}+2.5*\text{production amount}_{EEXE} \leq$

9600

Total production time (in minutes) of production line 2 should not exceed 9600 minutes

$1*\text{production amount}_{ESTU}+1*\text{production amount}_{ESTA}+1*\text{production amount}_{EEXE}+1*\text{production amount}_{BSTU}+1*\text{production amount}_{BSTA}+1*\text{production amount}_{BEXE} \leq 9600$

Total production time (in minutes) of production line 3 should not exceed 19200 minutes

$3*\text{production amount}_{BSTU}+4*\text{production amount}_{BSTA}+5*\text{production amount}_{BEXE}+3*\text{production amount}_{HCSTU}+4*\text{production amount}_{HCSTA}+5*\text{production amount}_{HCEXE} \leq 19200$

Total consumption for labor hour (in minutes) should not exceed 230400 minutes

$[2(1.5+1)+10]*\text{production amount}_{ESTU}+[2(2+1)+11]*\text{production amount}_{ESTA}+[2(2.5+1)+12]*\text{production amount}_{EEXE}+[2(1+4)+18]*\text{production amount}_{BSTA}+[2(1+5)+20]*\text{production amount}_{BEXE}+[23+20+50]*\text{production amount}_{HCSTU}+[24+25+60]*\text{production amount}_{HCSTA}+[25+30+70]*\text{production amount}_{HCEXE} \leq 230400$

3.3.3 Order Demand Constraints

The number of economy student desk produced in September should be larger or equal to the amount demanded

$\text{production amount}_{ESTU} \geq 750$

The number of economy standard desk produced in September should be larger or equal to the amount demanded

$\text{production amount}_{ESTA} \geq 1500$

The number of economy executive desk produced in September should be larger or equal to the amount demanded

$\text{production amount}_{EEXE} \geq 100$

The number of basic student desk produced in September should be larger or equal to the amount demanded

$\text{production amount}_{BSTU} \geq 400$

The number of basic standard desk produced in September should be larger or equal to the amount demanded

production amount$_{BSTA} \geq 1500$

The number of basic executive desk produced in September should be larger or equal to the amount demanded

production amount$_{BEXE} \geq 100$

The number of hand crafted student desk produced in September should be larger or equal to the amount demanded

production amount$_{HCSTU} \geq 25$

The number of hand crafted standard desk produced in September should be larger or equal to the amount demanded

production amount$_{HCSTA} \geq 150$

The number of hand crafted executive desk produced in September should be larger or equal to the amount demanded

production amount$_{HCEXE} \geq 50$

3.3.4 Quota Restrictions

Denote the total amount of desks produced by:

$\sum_{ESTU}^{HCSTA}$ prodcution amount = prodcution amount$_{ESTU}$+prodcution amount$_{ESTA}$+ prodcution amount$_{EEXE}$+prodcution amount$_{BSTU}$+prodcution amount$_{BSTA}$+ prodcution amount$_{BEXE}$+prodcution amount$_{HCSTU}$+prodcution amount$_{HCSTA}$+ prodcution amount$_{HCEXE}$

The amount of economy desk produced to the amount of all desks produced should conform to the range indicated by its quota

prodcution amount$_{ESTU}$+prodcution amount$_{ESTA}$+prodcution amount$_{EEXE} \geq 0.2 * \sum_{ESTU}^{HCSTA}$ prodcution amount

prodcution amount$_{ESTU}$+prodcution amount$_{ESTA}$+prodcution amount$_{EEXE} \leq 0.5 * \sum_{ESTU}^{HCSTA}$ prodcution amount

The amount of basic desk produced to the amount of all desks produced should conform to the range indicated by its quota

prodcution amount$_{BSTU}$+prodcution amount$_{BSTA}$+prodcution amount$_{BEXE} \geq$
$0.4 * \sum_{ESTU}^{HCSTA}$ prodcution amount

prodcution amount$_{BSTU}$+prodcution amount$_{BSTA}$+prodcution amount$_{BEXE} \leq$
$0.6 * \sum_{ESTU}^{HCSTA}$ prodcution amount

The amount of handcraft desk produced to the amount of all desks produced should conform to the range indicated by its quota

prodcution amount$_{HCSTU}$+prodcution amount$_{HCSTA}$+prodcution amount$_{HCEXE} \geq$
$0.1 * \sum_{ESTU}^{HCSTA}$ prodcution amount

prodcution amount$_{HCSTU}$+prodcution amount$_{HCSTA}$+prodcution amount$_{HCEXE} \leq$
$0.2 * \sum_{ESTU}^{HCSTA}$ prodcution amount

The amount of student desk produced to the amount of all desks produced should conform to the range indicated by its quota

prodcution amount$_{ESTU}$+prodcution amount$_{BSTU}$+prodcution amount$_{HCSTU} \geq$
$0.2 * \sum_{ESTU}^{HCSTA}$ prodcution amount

prodcution amount$_{ESTU}$+prodcution amount$_{BSTU}$+prodcution amount$_{HCSTU} \leq$
$0.35 * \sum_{ESTU}^{HCSTA}$ prodcution amount

The amount of standard desk produced to the amount of all desks produced should conform to the range indicated by its quota

prodcution amount$_{ESTA}$+prodcution amount$_{BSTA}$+prodcution amount$_{HCSTA} \geq$
$0.4 * \sum_{ESTU}^{HCSTA}$ prodcution amount

prodcution amount$_{ESTA}$+prodcution amount$_{BSTA}$+prodcution amount$_{HCSTA} \leq$
$0.7 * \sum_{ESTU}^{HCSTA}$ prodcution amount

The amount of executive desk produced to the amount of all desks produced should conform to the range indicated by its quota

prodcution amount$_{EEXE}$+prodcution amount$_{BEXE}$+prodcution amount$_{HCEXE} \geq$
$0.05 * \sum_{ESTU}^{HCSTA}$ prodcution amount

prodcution amount$_{EEXE}$+prodcution amount$_{BEXE}$+prodcution amount$_{HCEXE} \leq$
$0.15 * \sum_{ESTU}^{HCSTA}$ prodcution amount

Production Plan			
	Student Desk	Standard Desk	Executive Desk
Economy	750	1500	100
Basic	525	1657	825
Hand-crafted	25	1069	50

Table 1: Production Plan

4 Solution

We solved the problem using AMPL and the corresponding code was added into the Appendix. The optimal production plan for each type of the desks to reach the maximum profit are listed in the chart below(unit: desks)(Table 1). The non-integer answers were rounded down to their nearest integer value so satisfy the assumption we made in the early section. The corresponding bar chart is created and inserted below for a better visualization(Fig.1.):



Figure 1: Production Amount Distribution by Desk Type

The maximum profit the company could reach under the available resources and other requirements is $611,935. The calculation process is as followed:

$20*\text{production amount}_{ESTU}+30*\text{production amount}_{ESTA}+40*\text{production amount}_{EEXE}$
$+50*\text{production amount}_{BSTU}+80*\text{production amount}_{BSTA}+125*\text{production amount}_{BEXE}+$

$100*\text{production amount}_{HCSTU}+250*\text{production amount}_{HCSTA}+325*\text{production amount}_{HCEXE} = 20*750+30*1500+40*100+50*525+80*1657+125*825+100*25+250*1069+325*50 = \$611,935$

In order to reach this production goal, we offer a more detailed allocation schedule(Table 2) for each type of resources, production line and labor:

Resources Used						
Desk Type	Aluminum (square ft)	Particle board (square ft)	Pine Sheet (square ft)	Production Line1 (mins)	Production Line2 (mins)	Production Line3 (mins)
Economy Student	10500	6000	0	1125	750	0
Economy Standard	36000	22500	0	3000	1500	0
Economy Executive	3000	2400	0	250	100	0
Basic Student	0	0	11550	0	525	1575
Basic Standard	0	0	66280	0	1657	6628
Basic Executive	0	0	45375	0	825	4125
Hand-crafted Student	0	0	625	0	0	75
Hand-crafted Standard	0	0	48105	0	0	4276
Hand-crafted Executive	0	0	3000	0	0	250

Table 2: Resources Used

5 Discussion

In addition to the factors mentioned in the case description, there are some real-life conditions that are needed to be considered to deliver a more accurate and instructive production schedule.
This section discusses potential factors that would affect production but are not covered in the case description. Also, there is still room for improvement in some production variables in the current linear program model.

5.1 Potential Influencing Factors

5.1.1 Storage

Although the premise of this case is that all desks produced will be sold, there is still a time lag between production and sales, and that is where the inventory storage issue comes into being. For example, the placement of raw materials and desks are needed to be factored into the total cost of production. The manufacturer would be expected to adjust the timing and staffing if the cost of storage takes up a big portion of the total cost.

5.1.2 Supervisory Labor

In factories, in addition to the workforce that operates in front of the production line, supervisors are also part of the workforce components. Their job is to act as a bridge between managers and workers, ensuring that production schedules are accurately implemented and driven. They may not be a high percentage of the workforce, but including the cost of supervisory labor in the total cost of

production will bring the total cost of production optimized using the model set up closer to reality.

5.1.3 Transportation

Transportation is an important part of the post-production phase. During special period, transportation costs may increase due to factors like rising in crude oil prices, which ultimately translates into a significant increase in total costs. The inclusion of transportation costs in the production cost statistics also provides a more reasonable reference for the pricing system of the product.

5.2 Existing Parameter Settings

5.2.1 Onboarding Rate

The plant's current onboarding rate is 80%, indicating that there is still room for a relatively large available maximum workforce increase. Under the premise of ensuring legal rest hours, the factory can set up overtime pay and a personnel rotation system to increase available labor hours and improve productivity.

5.2.2 Production Line Setup

After adjusting the staffing schedule, the production tasks also need to be adjusted to achieve the ultimate goal of the adjustment, which is to improve cost efficiency. Whether more workforce enters the production line, in a specific line, such as line 3, for processing manual chairs, reflects more yield return is a subject that can be further investigated. The factory can take the chance speeding up its production process by rearranging work tasks to accomplish the same yield in a lower time-cost manner.

6 Sensitivity Analysis

Sensitivity Analysis of decision variables is presented in Table 3. The total profit is \$611,935, and optimal solutions are all nonzero, which means that every type of desks should be procuded to achieve this optimal profit under the current constraints. Table 3 also shows that the lower and upper bound which profit of each type of desks can vary to maintain the current optimal production schedule. Thus, Table 3 can be considered as a pricing guide, and we can ideally assume that if we can set the price of each type of desk to increase the profit to its upper bound without changing the cost, the maximum increase in profit would be \$98,351.80. The calculation is stated below:

$$\text{Increased Profit} = \sum \text{Final Value} * (\text{Upper Bound} - \text{Objective Coefficient})$$

Increased Profit = 750*(24.06-20) + 1500*(42.04-30) + 100*(78.33-40) +
525*(61.66-50) + 1657*(83.06-80) + 825*(171.25-125) + 25*(188.36-100) +

1069*(270.27-250) + 50*(328.76-325) = $98,351.80.
In addition, if the profit of each type of desk is less than its lower bound, the current optimal solution (production plan) will change. From the Table 3, we conclude that the profit of Basic-student desk cannot be lower than $45.3989; the profit of Basic-standard desk cannot be lower than $72.0293; the profit of Basic-executive desk cannot be lower than $121.219; the profit of Handcrafted-standard desk cannot be lower than $246.892.
Sensitivity Analysis of constraints is presented in Table 4. There are 28 constraints, and we divide the shadow price into 3 types: equal to zero, less than zero, and greater than zero. The shadow price of some constraints (production factors) equals to 0, indicating that changing values of these constraints (production factors) does not affect the current objective function value (the total profit). The shadow prices of some constraints (production factors) is negative. It indicates that increase in the value of these constraints would lead to decrease in the current objective function value (the total profit). For example, if the Economy-student order constraint increases by 1 unit, the total profit will decrease by $4.06. If the Economy-standard order constraint increases by 1 unit, the total profit will decrease by $12.05. If the Economy-executive order constraint increases by 1 unit, the total profit will decrease by $38.34. If the Handcraft-student order constraint increases by 1 unit, the total profit will decrease by $88.36. If the Handcraft -executive order constraint increases by 1 unit, the total profit will decrease by $3.77. If the Student quotas min constraint increase by 1%, the total profit will decrease by $0.12. Finally, the shadow price greater than 0 means increase in the value of these constraints would lead to increase in the current objective function value(the total profit). For example, if the total worker-minutes constraint changes from 230,400 to 230,401, the total profit will increase by $2.60. If the pine sheet constraint changes from 175,000 square feet to 175,001 square feet, the total profit will increase by $0.23. If 1% increase in maximum quotas of Executive desk, the total profit will increase by $0.31.
In the following text, we make a deep analysis on how the changes in availability of pine sheet, labor hours, and maximum quotas of Executive desk would affect CALDESCO's total profit respectively.

Variable Cells					
Desk Type	Final Value	Reduced Cost	Objective Coefficient	Lower Bound	Upper Bound
Economy Student	750	0	20	infinite	24.0623
Economy Standard	1500	0	30	infinite	42.0494
Economy Executive	100	0	40	infinite	78.3373
Basic Student	525.537	0	50	45.3989	61.6671
Basic Standard	1657.5	0	80	72.0293	83.0628
Basic Executive	825.403	0	125	121.219	171.259
Hand-crafted Student	25	0	100	infinite	188.363
Hand-crafted Standard	1069.24	0	250	246.892	270.172
Hand-crafted Executive	50	0	325	infinite	328.766

Table 3: Variable Cells

Constraints				
Name	Shadow Price	Constraint R.H Side	Lower Bound	Upper Bound
Aluminum	0	65000	49500	infinite
Particle Board	0	60000	30900	infinite
Pine Sheet	0.2345	175000	169646	195940
Line1	0	9600	4375	infinite
Line2	0	9600	5358.44	infinite
Line3	0	19200	16935,6	infinite
Total Labor Minutes	2.5974	230400	200443	239325
Economy Student Order	-4.0623	750	438.767	892.185
Economy Standard Order	-12.0494	1500	1187.61	2145.83
Economy Executive Order	-38.3373	100	0	616.667
Basic Student Order	0	400	infinite	525.537
Basic Standard Order	0	1500	infinite	1657.5
Basic Executive Order	0	100	infinite	825.403
Handcraft Student Order	-88.3629	25	0	151.282
Handcraft Standard Order	0	150	infinite	1069.24
Handcraft Executive Order	-3.7655	50	0	778.394
Economy Quotas Min	0	0	infinite	1049.46
Basic Quotas Min	0	0	infinite	407.367
Handcraft Quotas Min	0	0	infinite	493.976
Student Quotas Min	-12.7924	0	-114.494	281.915
Standard Quotas Min	0	0	infinite	1626.67
Executive Quotas Min	0	0	infinite	650.269
Economy Quotas Max	0	0	-901.343	infinite
Basic Quotas Max	0	0	-893.17	infinite
Handcraft Quotas Max	0	0	-893.17	infinite
Student Quotas Max	0	0	-975.403	infinite
Standard Quotas Max	0	0	-325.134	infinite
Executive Quotas Max	31.0932	0	-331.807	114.905

Table 4: Constrains

6.1 Pine Sheet Availability

The current available amount of pine sheets is 175,000 square feet. However, external factors that may affect the supply of pine sheets should also be considered.
The vast majority of pine trees grow in areas with arid conditions and thin soils. If the water content in the soil is too large, it will seriously affect the growth rate of pine trees. It would be reasonable to assume that yields of pine sheets may drop if there is a strong rainfall this year. If the available amount of pine sheet is reduced by a given amount, what will be the impact?

Suppose the available pine sheets become 150,000 square feet. The pine sheet availability constraint formula will be updated to:

$22*\text{production amount}_{BSTU}+40*\text{production amount}_{BSTA}+55*\text{production amount}_{BEXE}+$

Pine Sheet Constraint				
Name	Shadow Price	Constraint R.H. Side	Lower Bound	Upper Bound
Pine Sheet	0.2345	175000	169646	195940

Table 5: Pine Sheet Constraint

Production Plan After Changing Pine Sheet			
	Student Desk	Standard Desk	Executive Desk
Economy	750	1593	100
Basic	802	1540	100
Hand-crafted	25	480	716
Optimal Profit	597790		

Table 6: Production Plan After Changing Pine Sheet

25∗production amount$_{HCSTU}$+45∗production amount$_{HCSTA}$+60∗production amount$_{HCEXE} \leq$ 150000

However, the pine sheet has a lower bound restriction of 169646 square feet. The current optimal solution(production plan) will change as we change the pine sheet amount to a value lower than its lower bound. We run the AMPL model again but with a different right hand-side value and the result is showen in Table 6.

The new optimal total profit of \$597,790, and the revenue loss is \$611,935 - \$597,790= \$14,145 due to the shortage of the pine sheet.

6.2 Labor Hour Constraint

In our model, we followed the Case description and assumed that the CALDESCO employs a workforce of 30 craftsmen and only 80% of them are available during the month due to vacations, illnesses or other personal reasons. However, in real-life, there may a chance that more employees will be available throughout the month due to illness being a rare case or vacation policy varies. So in a real-life scenario, we may have more labor minutes available, which might lead to an increase on the right hand side of the total labour time constraint formula that exceeds the current upper bound of the labour time. Correspondingly, the output will change. It is worth to discuss how the total profit changes when the labor hour increases.

Suppose the total available labor minutes is increased to 250,000 minutes, which is above the upper bound of labour constraint. We run the program again and

Labor Hour Constraint				
Name	Shadow Price	Constraint R.H. Side	Lower Bound	Upper Bound
Total Labor Minutes	2.5974	0	200443	239325

Table 7: Labor Hour Constraint

Production Plan After Changing Labor Hour Constraint			
	Student Desk	Standard Desk	Executive Desk
Economy	886	1500	100
Basic	400	2161	198
Hand-crafted	25	601	684
Optimal Profit	659400		

Table 8: Production Plan After Changing Labor Hour Constraint

get the new production plan, which is shown in Table 8. The labor limit constraint formula will be updated to:

$15*\text{production amount}_{ESTU}+17*\text{production amount}_{ESTA}+19*\text{production amount}_{EEXE}+23*\text{production amount}_{BSTU}+28*\text{production amount}_{BSTA}+32*\text{production amount}_{BEXE}+76*\text{production amount}_{HCSTU}+93*\text{production amount}_{HCSTA}+110*\text{production amount}_{HCEXE} \leq 250000$

The total profit changed from \$611,935 to \$659,400. \$47,465 increment in profit occurs due to a 28800 minutes increase in labor minutes.

6.3 Quotas

The shadow price of executive quotas is \$31.0932, which means that if we produce 1% more executive desks, we would gain a \$31.0932 more than previous optimal total profit. Thus, producing more executive desks may lead to higher profits(if the marginal cost is not greater than the marginal revenue).

Suppose we assign a maximum quotas of 20% to produce executive desks.
The formula will be updated to:

$0.2*(\text{production amount}_{BSTU}+\text{production amount}_{BSTA}+\text{production amount}_{BEXE}+\text{production amount}_{ESTU}+\text{production amount}_{ESTA}+\text{production amount}_{EEXE}+\text{production amount}_{HCSTU}+\text{production amount}_{HCSTA}+\text{production amount}_{HCEXE}) \geq \text{production amount}_{ESTU}+\text{production amount}_{ESTA}+\text{production amount}_{EEXE}$

The new production plan is shown in Table 10.

Executive Quotas Max				
Name	Shadow Price	Constraint R.H. Side	Lower Bound	Upper Bound
Executive Quotas Max	31.0932	0	-331.807	114.905

Table 9: Executive Quotas Max

Production Plan After Changing Quotas			
	Student Desk	Standard Desk	Executive Desk
Economy	750	1500	100
Basic	501	1500	898
Hand-crafted	25	828	277
Optimal Profit	620825		

Table 10: Production Plan After Changing Quotas

The total profit changed from $611,935 to $620,825. $8,890 increment in total profit occurs due to the increase of the maximum limit of the executive products.

Appendices

7 AMPL Code

```
reset;
option solver cplex;
option cplex_options 'sensitivity primal';
option presolve  0;

# Sets
set VARS;
set RESOURCES;
set QUOTA_GROUP;

# parameters
# profit for each desk
param profit_info{VARS};
# September order for each type of desk
param sept_order_info{VARS};
# the amount of aluminum required for each type of desk
param aluminum_info{VARS};
# the amount of particle board required for each type of desk
param particle_board_info{VARS};
# the amount of pine sheet required for each type of desk
param pine_sheet_info{VARS};
# time needed to produce each type of desk at line1
param line1{VARS};
# time needed to produce each type of desk at line2
param line2{VARS};
# time needed to produce each type of desk at line3
param line3{VARS};
# assemble finshining time required for each type of desk
param AF{VARS};
# hand-crafting time required for each type of desk
param HC{VARS};

# resources availability for September
# labor
param LR;
# aluminum
param AL;
# particle board
param PB;
# pine sheets
param PS;
# production line 1
param PL1;
# production line 2
param PL2;
# production line 3
param PL3;

# production quotas
param minimum_perc{VARS, QUOTA_GROUP};
param maximum_perc{VARS, QUOTA_GROUP};
```

```
# Variables
# ESTU: economy student desk
# ESTA: economy standard desk
# EEXE: economy executive desk
# BSTU: basic student desk
# BSTA: basic standard desk
# BEXE: basic executive desk
# HCSTU: hand-crafted student desk
# HCSTA: hand-crafted standard desk
# HCEXE: hand-crafted executive desk
var production_amount{VARS} >= 0;

# Objective function
# We want to maximize the profit, which is computed by the sum of the amount of each desk produced times its corresponding profit
maximize total_profit:
    sum{desks in VARS} (production_amount[desks] * profit_info[desks]);

# Constraints
# material constrains
subject to aluminum_constraint:
    sum{desks in VARS} (production_amount[desks] * aluminum_info[desks]) <= AL;
subject to particle_board_constraint:
    sum{desks in VARS} (production_amount[desks] * particle_board_info[desks]) <= PB;
subject to pine_sheet_constrains:
    sum{desks in VARS} (production_amount[desks] * pine_sheet_info[desks]) <= PS;

# labor hour constraints
subject to line1_constraint:
    sum{desks in VARS} (production_amount[desks] * line1[desks]) <= PL1;
subject to line2_constraint:
    sum{desks in VARS} (production_amount[desks] * line2[desks]) <= PL2;
subject to line3_constraint:
    sum{desks in VARS} (production_amount[desks] * line3[desks]) <= PL3;
# from the article, we know the total amount of minimutes required to produce a desk equals to:
# 2 x (the total production line time) + (hand crafting time) + (assembly/finshing time)
subject to total_labor_hour_constrain:
    sum{desks in VARS} (production_amount[desks] * (2 * (line1[desks] + line2[desks] + line3[desks]) + AF[desks] + HC[desks])) <= LR;

# order demand constraint
subject to order_constraint{desks in VARS}:
    production_amount[desks] >= sept_order_info[desks];

# quotas constrains
subject to quotas_min{group in QUOTA_GROUP}:
    sum{desks in VARS} (minimum_perc[desks, group] * production_amount[desks]) >= 0;
subject to quotas_max{group in QUOTA_GROUP}:
    sum{desks in VARS} (maximum_perc[desks, group] * production_amount[desks]) <= 0;
```

```
set VARS := ESTU ESTA EEXE BSTU BSTA BEXE HCSTU HCSTA HCEXE;
set RESOURCES := LR AL PB PS PL1 PL2 PL3;
set QUOTA_GROUP := ECONOMY BASIC HANDC STUDENT STANDARD EXECUTIVE;

param profit_info :=
    ESTU    20
    ESTA    30
    EEXE    40
    BSTU    50
    BSTA    80
    BEXE    125
    HCSTU   100
    HCSTA   250
    HCEXE   325;

param sept_order_info :=
    ESTU    750
    ESTA    1500
    EEXE    100
    BSTU    400
    BSTA    1500
    BEXE    100
    HCSTU   25
    HCSTA   150
    HCEXE   50;

param aluminum_info :=
    ESTU    14
    ESTA    24
    EEXE    30
    BSTU    0
    BSTA    0
    BEXE    0
    HCSTU   0
    HCSTA   0
    HCEXE   0;

param particle_board_info :=
    ESTU    8
    ESTA    15
    EEXE    24
    BSTU    0
    BSTA    0
    BEXE    0
    HCSTU   0
    HCSTA   0
    HCEXE   0;

param pine_sheet_info :=
    ESTU    0
    ESTA    0
    EEXE    0
    BSTU    22
    BSTA    40
    BEXE    55
    HCSTU   25
    HCSTA   45
    HCEXE   60;
```

```
param line1 :=
    ESTU    1.5
    ESTA    2
    EEXE    2.5
    BSTU    0
    BSTA    0
    BEXE    0
    HCSTU   0
    HCSTA   0
    HCEXE   0;

param line2 :=
    ESTU    1
    ESTA    1
    EEXE    1
    BSTU    1
    BSTA    1
    BEXE    1
    HCSTU   0
    HCSTA   0
    HCEXE   0;

param line3 :=
    ESTU    0
    ESTA    0
    EEXE    0
    BSTU    3
    BSTA    4
    BEXE    5
    HCSTU   3
    HCSTA   4
    HCEXE   5;

param AF :=
    ESTU    10
    ESTA    11
    EEXE    12
    BSTU    15
    BSTA    18
    BEXE    20
    HCSTU   20
    HCSTA   25
    HCEXE   30;

param HC :=
    ESTU    0
    ESTA    0
    EEXE    0
    BSTU    0
    BSTA    0
    BEXE    0
    HCSTU   50
    HCSTA   60
    HCEXE   70;
```

```
param LR := 230400;
param AL := 65000;
param PB := 60000;
param PS := 175000;
param PL1 := 9600;
param PL2 := 9600;
param PL3 := 19200;

param minimum_perc :
           ECONOMY     BASIC       HANDC       STUDENT     STANDARD        EXECUTIVE :=
    ESTU   0.8         -0.4        -0.1        0.8         -0.4            -0.05
    ESTA   0.8         -0.4        -0.1        -0.2        0.6             -0.05
    EEXE   0.8         -0.4        -0.1        -0.2        -0.4            0.95
    BSTU   -0.2        0.6         -0.1        0.8         -0.4            -0.05
    BSTA   -0.2        0.6         -0.1        -0.2        0.6             -0.05
    BEXE   -0.2        0.6         -0.1        -0.2        -0.4            0.95
    HCSTU  -0.2        -0.4        0.9         0.8         -0.4            -0.05
    HCSTA  -0.2        -0.4        0.9         -0.2        0.6             -0.05
    HCEXE  -0.2        -0.4        0.9         -0.2        -0.4            0.95;

param maximum_perc :
           ECONOMY     BASIC       HANDC       STUDENT     STANDARD        EXECUTIVE :=
    ESTU   0.5         -0.6        -0.2        0.65        -0.7            -0.15
    ESTA   0.5         -0.6        -0.2        -0.35       0.3             -0.15
    EEXE   0.5         -0.6        -0.2        -0.35       -0.7            0.85
    BSTU   -0.5        0.4         -0.2        0.65        -0.7            -0.15
    BSTA   -0.5        0.4         -0.2        -0.35       0.3             -0.15
    BEXE   -0.5        0.4         -0.2        -0.35       -0.7            0.85
    HCSTU  -0.5        -0.6        0.8         0.65        -0.7            -0.15
    HCSTA  -0.5        -0.6        0.8         -0.35       0.3             -0.15
    HCEXE  -0.5        -0.6        0.8         -0.35       -0.7            0.85;
```